
OKVIR: VBA Makro za automatsko numeriranje jednadžbi

Public Sub MAIN()
Dim vfc
Dim CurrentHeading$

' Zamrzava sliku i pamti trenutno stanje prikaza kodova
Application.ScreenUpdating = False
vfc = ActiveWindow.View.ShowFieldCodes

' Primjenjuje stil, umeće tabulatore i oznaku formule
Selection.Style = "formula"
Selection.TypeText Text:=Chr(9) + Chr(9) + "("

' Umeće referencu na tekući naslov
Selection.Fields.Add Range:=Selection.Range, Type:=wdFieldStyleRef,_ 
  Text:="1 \n"
ActiveWindow.View.ShowFieldCodes = False
Selection.MoveLeft Unit:=wdWord, Count:=1, Extend:=wdExtend
CurrentHeading$ = Selection.Text
Selection.MoveRight Unit:=wdCharacter, Count:=1, Extend:=wdMove
Selection.TypeText Text:="."

ActiveDocument.Bookmarks.Add Name:="currentPosition"

' Umeće SEQ polje. Ako je prva jednadžba u nizu, tada resetira brojač na 1
If EquationBefore(CurrentHeading$) = True Then
    ActiveDocument.Fields.Add Range:=Selection.Range, _
      Type:=wdFieldSequence, Text:="Equation \* ARABIC"
Else
    ActiveDocument.Fields.Add Range:=Selection.Range, _
      Type:=wdFieldSequence, Text:="Equation \* ARABIC \r 1"
End If
Selection.TypeText Text:=")"

' Provjerava ima li jednadžbi iza
CheckNextEquation (CurrentHeading$)

ActiveDocument.Bookmarks("currentPosition").Delete

' Umeće jednadžbu (editor jednadžbi mora biti instaliran!)
Selection.HomeKey Unit:=wdLine, Extend:=wdMove
Selection.MoveRight Unit:=wdCharacter, Count:=1, Extend:=wdMove
ActiveWindow.View.ShowFieldCodes = vfc
Application.ScreenUpdating = True
Selection.InlineShapes.AddOLEObject ClassType:="Equation.3", FileName:="",_
  LinkToFile:=False, DisplayAsIcon:=False

End Sub

Private Function EquationBefore(CurrHead$)

ActiveWindow.View.ShowFieldCodes = True

With Selection.Find
    .ClearFormatting
    .Style = "formula"
    .Execute FindText:="STYLEREF 1 \n", Format:=True, Forward:=False
    If .Found = False Then



        EquationBefore = False
    Else
        ActiveWindow.View.ShowFieldCodes = 0
        Selection.MoveRight Unit:=wdWord, Count:=1, Extend:=wdExtend
        If CurrHead$ = Selection.Text Then
            EquationBefore = True
        Else
            EquationBefore = False
        End If
    End If
End With

Selection.GoTo What:=wdGoToBookmark, Name:="currentPosition"
ActiveWindow.View.ShowFieldCodes = False

End Function

Private Sub CheckNextEquation(CurrentHeading$)

ActiveWindow.View.ShowFieldCodes = True

With Selection.Find
    .ClearFormatting
    .Style = "formula"
    .Execute FindText:="SEQ Equation \* ARABIC \r 1", Format:=True, _
      Forward:=True
    If .Found = True Then
        Selection.MoveLeft Unit:=wdCharacter, Count:=1, Extend:=wdMove
        Selection.Find.Execute FindText:="STYLEREF 1 \n", Format:=True, _
          Forward:=False
        ActiveWindow.View.ShowFieldCodes = False
        Selection.MoveRight Unit:=wdWord, Count:=1, Extend:=wdExtend
        Selection.Fields.Update
        If CurrentHeading$ = Selection.Text Then
            ActiveWindow.View.ShowFieldCodes = True
            Selection.MoveLeft Unit:=wdCharacter, Count:=1, Extend:=wdMove
            With Selection.Find
                .Execute FindText:=" \r 1", Forward:=True, Format:=True
            End With
            Selection.Delete
        End If
    End If
End With

ActiveWindow.View.ShowFieldCodes = False
Selection.GoTo What:=wdGoToBookmark, Name:="currentPosition"

End Sub


